Journal of Financial Education
Fall 1988

A CALCULATOR PROGRAM FOR OPTION VALUES AND IMPLIED STANDARD DEVIATIONS

Peter Carr

Assistant Professor of Finance, Cornell University

Taylor [2] implemented the Black Scholes [1] call option pricing model on the HP12C, a programmable financial calculator[1]. This paper presents an alternative program, that is easier to use while providing more outputs, such as put values and implied standard deviations.

The popularity of the Black Scholes model is principally due to the fact that most of its inputs are observable. The main unobservable variable is the instantaneous standard deviation of the underlying asset's rate of return. The weight of academic research suggests that this variable is best estimated from market prices using the Black Scholes formula. The program in this paper directly determines this implied standard deviation (isd) using a Newton-Raphson scheme. The isd can then be used to calculate the model price of other European calls and puts on the same underlying asset. The sensitivity of these option values to the underlying asset's price and return volatility are also generated.

The main advantage of the program in this paper over that presented in Taylor [2] is that more outputs are calculated. Taylor's program calculates a single call value, along with its hedge ratio and that of the corresponding European put. In addition, to these quantities, the current program also calculates the European put's value, volatility derivatives and the implied standard deviation. Furthermore, the user interface in the current program is friendlier. The five inputs to the Black Scholes formula are simply stored in the calculator and the program is run. In contrast, Taylor's program requires that some ancillary values be stored in addition. Both programs are of the same length, which is the maximum length allowed in the HP12C.

The remaining five sections of the paper correspond to the five steps required to implement the program. A running example is employed to facilitate this implementation.

STEP 1 - ENTER THE PROGRAM IN THE CALCULATOR

The following program, Table 1, need be entered only once[2]. It remains

---

[1]Although handheld computers are fast approaching calculators in portability and price, many calculators will continue to be used. Hopefully, this program will allow the venerable HP12C to compete with these new computers.

[2]The approximation to the standard normal distribution function slightly modifies one given in the HP12C solutions handbook. The approximation is:

$$N(x) = 0.5*\exp\{-0.5*[x^2 + 562/(83-351/x)]\}, \quad \text{if } x < 0$$
$$= 1 - 0.5*\exp\{-0.5*[x^2 + 562/(83+351/x)]\}, \quad \text{if } x >= 0$$

TABLE 1

HP12C OPTION PROGRAM

Press	Display	Press	Display	Press	Display
RCL FV	01- 45 15	RCL 1	34- 45 1	$x \gtrless y$	67- 34
STO 6	02- 44 6	STO 4	35- 44 4	STO 2	68- 44 2
g INTG	03- 43 25	STO X 4	36- 44 20 4	g $x \leq y$	69- 43 34
.	04- 48	2	37- 2	g GTO 34	70- 43,33 34
4	05- 4	8	38- 8	STO 4	71- 44 4
STO 0	06- 44 0	1	39- 1	RCL PV	72- 45 13
g $x \leq y$	07- 43 34	CHS	40- 16	X	73- 20
STO 6	08- 44 6	ENTER	41- 36	$x \gtrless y$	74- 34
RCL PV	09- 45 13	3	42- 3	RCL 3	75- 45 3
STO X 0	10- 44 20 0	5	43- 5	X	76- 20
RCL PMT	11- 45 14	1	44- 1	-	77- 30
RCL n	12- 45 11	RCL 4	45- 45 4	STO 1	78- 44 1
RCL i	13- 45 12	g y^x	46- 43 21	RCL PV	79- 45 13
X	14- 20	÷	47- 10	-	80- 30
g e^x	15- 43 22	8	48- 8	STO + 3	81- 44 40 3
÷	16- 10	3	49- 3	RCL 5	82- 45 5
STO 3	17- 44 3	+	50- 40	STO ÷ 0	83- 44 10 0
÷	18- 10	÷	51- 10	1	84- 1
g LN	19- 43 23	g e^x	52- 43 22	STO - 4	85- 44 30 4
RCL 6	20- 45 6	RCL 4	53- 45 4	RCL FV	86- 45 15
RCL n	21- 45 11	g e^x	54- 43 22	g $x \leq y$	87- 43 34
g y^x	22- 43 21	g y^x	55- 43 21	g GTO 99	88- 43,33 99
STO X 0	23- 44 20 0	STO 5	56- 44 5	RCL 1	89- 45 1
X	24- 20	÷	57- 10	-	90- 30
STO 1	25- 44 1	2	58- 2	RCL 0	91- 45 0
÷	26- 10	÷	59- 10	÷	92- 10
g LST X	27- 43 36	g INTG	60- 43 25	RCL 6	93- 45 6
2	28- 2	g $x \leq y$	61- 43 34	+	94- 40
STO 2	29- 44 2	1	62- 1	STO 6	95- 44 6
÷	30- 10	g LST X	63- 43 36	R/S	96- 31
-	31- 30	g $x \leq y$	64- 43 34	RCL 6	97- 45 6
STO + 1	32- 44 40 1	-	65- 30	g GTO 03	98- 43,33 03
g GTO 35	33- 43,33 35	RCL 2	66- 45 2	RCL 1	99- 45 1

in memory even after the calculator is turned off, unless the program memory is specifically cleared. To enter the program, turn the calculator on and toggle it into "program mode" by pressing f P/R[3]. Table 1 indicates the

[3]This is done by first hitting the gold f key and then pressing the R/S key. This keystroke combination toggles the P/R switch embossed in gold over the R/S key (P/R stands for Program mode/Run mode). The calculator should now be in program mode as indicated by the word PRGM in the lower right corner of the display. The 00- on the left of the display indicates that the current line number is zero.

program lines that must be entered. To reduce the possibility for keypunch error, the table also shows the calculator display after each program line has been entered. It should take about five minutes to enter all 99 lines. When finished, press f P/R to toggle the calculator back into run mode.

STEP 2 - INPUT OPTION PARAMETERS

The Black Scholes formula depends on four observable variables:

n, the number of years to maturity, e.g., 0.33 years.
i, the interest rate as a decimal, e.g., .12 per year.
PV, the present value of the underlying asset, i.e., the current stock price,[4] e.g., $40.
PMT, the payment required to exercise the option, i.e., the option strike price, e.g., $40.

Each parameter is stored in the financial register of the same name. To store a parameter, press the number and then the corresponding register key. Thus, for this examples, press:

.33	n
.12	i
40	PV
40	PMT

The user must store a fifth value in the last financial register FV. At this point, the user must decide whether the program is to calculate an implied standard deviation or an option price. If an implied standard deviation is desired, the market call price is the final value and is stored in FV. Conversely, if an option value is desired, the annualized standard deviation is the fifth value and is stored in FV. The program distinguishes between the two alternative uses by the magnitude of the number stored in FV. If this number is less than one, then the program assumes it is a standard deviation and calculates option values. Conversely, if the number is equal to or greater than one, then the program assumes it is a market call price and calculates its isd. Because in practice, almost all standard deviations are less than one hundred percent and almost all market call prices exceed one dollar, this feature should not present any difficulties[5].

---

[4]If the stock pays dividends, enter the stock price less the present value of all dividends paid to the option's maturity. If the option is on the futures price, enter the futures price times the discount factor for the option's maturity. Note that the program only calculates European option prices and that these may differ from American option prices when the stock pays a dividend or the option is written on a futures price.

[5]However, market call prices less than one dollar can be handled by multiplying the stock, strike and call prices entered by 100. This will not change the isd calculated. Similarly, if the standard deviation is greater than 100 percent but less than one thousand percent, divide it by 10 and multiply the time to maturity so that the product of the standard deviation and the square root of maturity is unchanged. Also divide the

To continue the example, suppose that the volatility is unknown so that the user intends to imply it from the market price of $3.54. In this case, the fifth value is stored in the last financial register by pressing 3.54 and then FV[6].

STEP 3. RUN THE PROGRAM

Press R/S to run the program. The display will blink "running" for about ten seconds and then a number will appear. If a standard deviation was entered in FV, then this number is the Black Scholes call price. Conversely, if a market call price was entered in FV, then this number is the first approximation to the true isd. In either case, various outputs can be retrieved as detailed in the next section. If the approximation to the true isd is in the display, then pressing R/S again will calculate a better approximation. The user should continue iterating until successive approximations are sufficiently close to each other.

For the running example, the first approximation to the isd is 0.3002. Pressing R/S again leads to a closer approximation of 0.2999. Because the two approximations are so close, no further iterations should be required[7].

STEP 4. REVIEW VARIOUS OUTPUTS

The program calculates European call and put prices, along with their hedge ratios and sensitivities to return volatility. Table 2 indicates where these outputs can be found. To retrieve an output, press RCL and the associated storage register number. The example values given in Table 2 assume two iterations as described above.

In deciding when to stop iterating for an isd, one can keep reviewing the model call value until it is sufficiently close to the market price. Note that outputs are based on the previous isd and not the latest one calculated. For example, model values shown in Table 2 correspond to the first approximation of 0.3002 and not the second approximation of 0.2999. Because the model call value of $3.54 is within half a cent of the market price, it is clear that the true isd is around 0.3 and that two iterations were sufficient in this example. In general, the number of iterations required depends on the sensitivity of an option to the underlying asset's volatility. It is easy to show that this derivative is the same for calls and puts, using put-call parity. Thus, a .01 rise in standard deviation would raise both of the above call and put prices by about 8 cents.

---

interest rate by 100 so that the product of the interest rate and the time to maturity is unchanged.

[6]If the user wants to calculate option values directly, then press 0.3, then FV where the annual standard deviation is 30 percent.

[7]If the numbers obtained disagree with the example given, check that the option parameters were entered correctly by recalling the financial registers and that the program was entered correctly by pressing SST in the run mode and comparing the display against the program in Table 1.

TABLE 2

LOCATION OF OUTPUTS

Output	Storage Register	Values in Example
Call price	1	$3.54
Call hedge ratio	2	.624
Put price	3	$1.99
Put hedge ratio	4	-.376
Derivative of call price with respect to standard deviation	0	8.74

STEP 5. SENSITIVITY ANALYSIS

Armed with an implied standard deviation of 0.3, the model next calculates values for other options on the same underlying asset. Suppose one wishes to calculate the value of an otherwise identical call option, but with an exercise price of $35. Because the financial registers are not altered by the execution of the program, one need only press:

```
35    PMT
0.3    FV
      R/S
```

The call and put values are $6.87 and $0.52, respectively. Their common sensitivity to volatility is 5.068. Thus, a .01 increase in the standard deviation of 0.3 should increase both call and put option values by about 5 cents. To test this, store .31 in FV and run the program by pressing R/S. The call and put option values in storage registers 1 and 3 are 5 cents higher as expected.

BIBLIOGRAPHY

1. Black, Fischer and Myron Scholes, "The Pricing of Options and Corporate Liabilities," Journal of Political Economy, pp. 637-54.

2. Taylor, Richard W. "Option Valuation for Alternative Instruments with the Black-Scholes Model: A Pedagogical Note," Journal of Financial Education, Fall 1987, pp. 73-77.